
2 Correction to Supplimentary Service testcase 15.8.5
2.1 f_TC_15_8_5_UTRAN() 
	Object name 
	f_TC_15_8_5_UTRAN()

	Reason for change
	The current TTCN implementation doesn’t take into account that a UE may use different TIs for call related and non-call related procedures. This needs to be accommodated.

	Summary of change
	Created four new variables v_TI_R_CC, v_TI_S_CC, v_TI_R_SS and v_TI_S_SS to handle the possibility that a UE may use different TIs for call related and non-call related procedures.

	Module
	CallRestriction.ttcn

	Label
	WA#


Before:
	function f_TC_15_8_5_UTRAN() runs on UTRAN_PTC

  { /* Call restriction - Rejection after invoke of ActivateSS operation */

    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;

    var Components v_FacilityInvokeBarring;

    var octetstring v_FacilityReturn;

    var TI v_TI_R;

    var TI v_TI_S;  

    //Initialise all UTRAN cells according to 34.108

    f_UTRAN34_Init();

    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);

    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);

    // The UE is in call state U10 by using table 10.1.2/3

    v_TI_R := f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); //Establishment of an outgoing call TS34.123, Table 10.1.2/3

    f_UTRAN_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // The UE is made to initiate an activation of BOIC (Barring of outgoing international calls) supplementary service

    f_UT_SupplementaryService_CallBarring(UT, //@sic R5s150893 sic@

                                          ACTIVATE_MODE,

                                          SS_BOIC);

    // @siclog "Steps 2-4" siclog@

    // CM SERVICE REQUEST - REGISTER //@sic R5s141087 R5-153702 sic@

    v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeActivate(?, cr_SS_ForBS_Code(tsc_SS_Code_BOIC, *)),

                                                              "Step 2",

                                                              "Step 4");

    v_FacilityInvokeBarring := v_ReceiveRegisterResult.components;

    v_TI_R := v_ReceiveRegisterResult.ti;

    // Set TI value for response to be sent from the SS

    v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);

    // Assign invokeID in the ReturnError with the value received in the Invoke

    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode(v_FacilityInvokeBarring[0].invoke.invokeID, tsc_ErrorCode_SS_SubscriptionViolation))));

    // @siclog "Step 5" siclog@

    // SS sends RELEASE COMPLETE with Return error component with SS_SubscriptionViolation

    U_Dc.send (cas_DataReq_CS (utran_CellDedicated,

                               tsc_RB3,

                               cs_U_ReleaseCompleteFacility (v_TI_S, cs_FacilityIE_tlv(v_FacilityReturn))));

    // @siclog "Step 6" siclog@

    // The UE provides an appropriate user indication about failure of the supplementary service activation

    if (f_UT_ChkSuppServiceUserIndication(UT, "SS_BOIC", "Password registration failure due to subscription violation")) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    // @siclog "Step 7-8" siclog@

    f_UTRAN34_CheckStatus(v_TI_S, v_TI_R, "Step 8", -, tsc_CallState_U10);

    //Postamble

    f_UTRAN_TestBody_Set(false);

    f_UTRAN_CC_Disconnect(utran34_Cell1, v_TI_S);

    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);

    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType (utran34_Cell1));

  } // end of f_TC_15_8_5_UTRAN


After:
	function f_TC_15_8_5_UTRAN() runs on UTRAN_PTC

  { /* Call restriction - Rejection after invoke of ActivateSS operation */

    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;

    var Components v_FacilityInvokeBarring;

    var octetstring v_FacilityReturn;

    var TI v_TI_R_CC; //WA#R5s160xxx: introduced additional variables to consider the possibility that the UE may use different TI values for call related and non call related supplementary services

    var TI v_TI_S_CC;

    var TI v_TI_R_SS;

    var TI v_TI_S_SS;
    //Initialise all UTRAN cells according to 34.108

    f_UTRAN34_Init();

    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);

    f_UTRAN34_SendDefSysInfo(utran34_Cell1);

    f_UTRAN34_Preamble(utran34_Cell1);

    // The UE is in call state U10 by using table 10.1.2/3

    //WA#R5s160xxx
    v_TI_R_CC := f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); //Establishment of an outgoing call TS34.123, Table 10.1.2/3

    f_UTRAN_TestBody_Set(true);

    // @siclog "Step 1" siclog@

    // The UE is made to initiate an activation of BOIC (Barring of outgoing international calls) supplementary service

    f_UT_SupplementaryService_CallBarring(UT, //@sic R5s150893 sic@

                                          ACTIVATE_MODE,

                                          SS_BOIC);

    // @siclog "Steps 2-4" siclog@

    // CM SERVICE REQUEST - REGISTER //@sic R5s141087 R5-153702 sic@

    v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeActivate(?, cr_SS_ForBS_Code(tsc_SS_Code_BOIC, *)),

                                                              "Step 2",

                                                              "Step 4");

    v_FacilityInvokeBarring := v_ReceiveRegisterResult.components;

    v_TI_R_SS := v_ReceiveRegisterResult.ti; //WA#R5s160xxx
    // Set TI value for response to be sent from the SS

    v_TI_S_CC := f_NAS_TI_InitAndSetFlag(v_TI_R_CC, '1'B); //WA#R5s160xxx
    v_TI_S_SS := f_NAS_TI_InitAndSetFlag(v_TI_R_SS, '1'B); //WA#R5s160xxx
    // Assign invokeID in the ReturnError with the value received in the Invoke

    v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode(v_FacilityInvokeBarring[0].invoke.invokeID, tsc_ErrorCode_SS_SubscriptionViolation))));

    // @siclog "Step 5" siclog@

    // SS sends RELEASE COMPLETE with Return error component with SS_SubscriptionViolation

    U_Dc.send (cas_DataReq_CS (utran_CellDedicated,

                               tsc_RB3,

                               cs_U_ReleaseCompleteFacility (v_TI_S_SS, cs_FacilityIE_tlv(v_FacilityReturn)))); //WA#R5s160xxx
    // @siclog "Step 6" siclog@

    // The UE provides an appropriate user indication about failure of the supplementary service activation

    if (f_UT_ChkSuppServiceUserIndication(UT, "SS_BOIC", "Password registration failure due to subscription violation")) {

      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    } else {

      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");

    }

    // @siclog "Step 7-8" siclog@

    f_UTRAN34_CheckStatus(v_TI_S_CC, v_TI_R_CC, "Step 8", -, tsc_CallState_U10); //WA#R5s160xxx
    //Postamble

    f_UTRAN_TestBody_Set(false);

    f_UTRAN_CC_Disconnect(utran34_Cell1, v_TI_S_CC); //WA#R5s160xxx
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);

    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType (utran34_Cell1));

  } // end of f_TC_15_8_5_UTRAN


	

	


